ルンゲ・クッタ法とマトリクスを用いたロレンツの非線形連立微分方程式の計算

SHIMURA Masato
jcd02773@nifty.com
URL:http://homepage3.nifty.com/asagaya_avenue

2014 年 6 月 12 日

目次

はじめに

runge-kutta 法 (RK4、４次) に関し、J 言語で関数（動詞）を引数として用いるシンプルなスクリプトを「ロレンツの微分方程式の線形計算」（JAPLA 2014 年 1 月）として発表したが、更にその応用を図る。

1 ルンゲ・クッタ法

微分方程式が用いられるようになって 300 年ほどが経ち、数々の解法が考案されている。ルンゲ・クッタ法は 1901 年に考案され、常微分方程式の数値解法の定番と位置づけられているが、当時はコンピュータなどなかった。

色々な言語の数学関数などを比較したロゼッタ・コード

`rosettacode.org/wiki/Runge-Kutta_method`

に 30 近い各種の言語で書かれたルンゲ・クッタ法のコードが載っている。中には `PL/I`

や Python,JavaScript なども あって見ていて楽しいが、アルゴリズム上繰り返しを用いなければならないため各言語による差異は少なく、Mathematica も数式を入れて $\cdots$ とは行かず、苦戦している。その中に誰かが書いた J のスクリプトもあがっているが洗練されているとはいい難い。

1.1 Runge と Kutta のプロフィール

Carl D.T. Runge（*1857-1927)* は父の仕事の関係でキューバのハバナで生まれた。ベルリン大学ではワイエル・シュトラウス門下。晩年はクラインの招聘でゲッチンゲン大学へ移った。

Marti. W. Kutta(1867-1944) はドイツ領だったシレジア地方の生まれで、ブラスラウやミュンヘンで学び、ケンブリッジに留学した後、アーヘンやシュツットガルトの大学で教鞭をとった。

1.2 C.Reiter の方法

C.Reiter の「*Fractal visualization and J* 第 *3* 版」でマトリクスとルンゲクッタ法でこの微分方程式を計算しているのを見受けた。マトリクス計算の好例なのでフォローしてみよう。

ルンゲクッタ法はロレンツの微分方程式自体を *(*左）引数とするので副詞型で作成する。

1. *C.Reiter* によるルンゲクッタ法のスクリプト。ループは外付けなので、実にシンプルである。

```
   rk=:1 : 0
:
h2=. -: x
k1=. u y
k2=. u y + h2*k1
k3=. u y + h2*k2
k4=. u y + x *k3
y+(x%6)*k1+k4++:k2+k3
)
```

2. スクリプトの解説

- *1 : 0* は動詞 *(*関数*)* を引数に取る副詞型
- *u* は関数（動詞）を左引数として入力
- *-: halve 1r2* $\frac{1}{2}$
- *+: double*= $\times 2$

2　幾つかの微分方程式

非線形連立微分方程式のロレンツとロスラーモデルは馴染んでいるので最初の例題として取り上げよう

2.1　ロレンツモデル

ロレンツは下から暖められ、上で冷える大気の流体モデルを考案し、無限個の変数を含む微分方程式体系を大胆に *3* 個の変数以外は定数であるとするモデル単純化を行った。一つは滞留運動の速度 *(x)* で*,* 他の *2* は水平方向と垂直方向の温度変異 *(y* と *z)* を測るものとなった。

$$\begin{cases} X' = & -\sigma X & +\sigma Y & \\ Y' = & -XZ & +\gamma X & -Y \\ Z' = & XY & -bZ & \end{cases}$$

[*1]

そしてこの *3* 個のパラメータは σ がプランドル数、γ がレイリー数、b が系の物理サイズに係わるものである。

$\sigma = 10, b = \frac{3}{8}, \gamma = 28$ がロレンツのオリジナル · パラメータである。

2.1.1　線形化

ロレンツの連立微分方程式は線形化すると次のような系になる。（*Hirsh p.312)*

$$Y' = \begin{pmatrix} -\sigma & \sigma & 0 \\ \gamma - z & -1 & -x \\ y & x & -b \end{pmatrix} Y$$

[*1] ロレンツモデルは非圧縮条件、ナビエ · ストークス方程式、熱伝導方程式の 3 式から導かれる。導出過程は [松葉] に詳しい

2.1.2 C.Reiter の lz の script

1. ロレンツの微分方程式のスクリプト

   C.Reiter のロレンツのマトリクス。横長のマトリクスを用いて巧妙に作成されている

```
NB. Lorenz
lz=: 1 : 0
'S B R'=. m                                        NB. s b r
M=. ((-S),S,0 0 0),(R,_1 0 0 _1),: 0 0 ,(-B),1 0   NB. make matrix
M&(+/ . *)@(] , {. * }.)                           NB. 計算部分
)
```

   (*a*) `+/ . *` は内積計算を行う。`M (+/ . *)` **初期値**

   (*b*) 初期値の数はマトリクス *M* の列の数と等しくなければならない

   (*c*) 出力は *M* の行数すなわち連立方程式の元の数となる。

   (*d*) ループ `^:(i. n)` では *(c)* の出力を *1* 回づつ受けて、xz, yz をその都度構成して入力する

2. パラメータをマトリクスに当てはめると

   $\sigma = 10, b = \dfrac{3}{8}, \gamma = 28$ がロレンツのオリジナル · パラメータ

$$\begin{cases} X' = & -\sigma X & +\sigma Y & & & \\ Y' = & \gamma X & -Y & & & -XZ \\ Z' = & & & -bZ & +XY & \end{cases}$$

	x	y	z	xy	xz
X'	-10	10	0	0	0
Y'	28	-1	0	0	-1
Z'	0	0	$-8r3$	1	0

```
   M
_10 10    0 0  0
 28 _1    0 0 _1
  0  0 _8r3 1  0
```

$Y' = \gamma X - Y - XZ$ の $-XZ$ と $Z' = XY - bZ$ の XY を別項に書き出している。ネスティッド · アレー（*2* 重配列）や文字列操作なども不要で、非線形方程式も横長マ

トリクスで計算できる。

3. x, y, z, xy, xz に入る値。（初期値とリピート入力）　x, y, z と xy, xz の作成

```
   (] , {. * }.) 0.1 0.2 0.3
0.1 0.2 0.3 0.02 0.03
```

（a）*] 0.1 0.2 0.3 → 0.1 0.2 0.3*

（b）*xy xz* の作成　`{. *  }.`　*→ 0.1 * 0.2 0.3*

4. *run*（数値計算）

```
 0.002 (10 8r3 28) lz rk ^:(i.100) 0.1 0.2 0.3
```

5. *Usage: (plot)*

```
 'noaxes' plot {@|:   0.002 (10 8r3 28 lz) rk ^:(i.30000) 0.1 0.2 0.3
```



2.2 ロスラーモデル

マックス・プランク研究所発のロレンツモデルを簡素化したカオスモデル。

1. 式はマトリクスに組むため順序を整理してある。b は定数

$$\begin{cases} \frac{dx}{dt} = & & -y & -z & & \\ \frac{dy}{dt} = & x & +ay & & & \\ \frac{dz}{dt} = & & & -cz & +xz & +b \end{cases}$$

2. ロスラーのスクリプト

```
NB. Rossler
ros=: 1 : 0
'A B C'=. m                                                      NB. a b c
M=: (0 _1 _1 0 0 ),(1,(A), 0 0 0),: 0 0 ,(-C),1,(B)   NB. make matrix
M&(+/ . *)@(] , ({. * {:),1:)                                   NB. 計算部分
)
```

(*a*) `{. * {:` xz の部分

(*b*) `{:` 最後尾を採る

(*c*) *1:* *1* を出力する動詞。 ここでは定数を取り出している

3. マトリクス

	x	y	z	xz	T
X'	0	−1	−1	0	0
Y'	1	0.2	0	0	0
Z'	0	0	−5.6	1	0.2

```
    M
0  _1    _1 0   0
1 0.2     0 0   0
0   0 _5.6 1 0.2
```

4. パラメータと初期値の例

```
   0.2 0.2 5.6 ros 1 0 0
0 1 1
```

5. *Usage:*

```
 'noaxes 'plot {@|:   0.002( 0.2 0.2 5.6)  ros rk ^:(i.60000)  1 0 0
```



References

Cliff Reiter [Fractal Visualization and J] 3rd edition Lulu 2007

M.Hirsch S.Smale R.L.Devaney 「力学系入門　原著　第 *2* 版」共立出版　*2007*

松葉育雄　「複雑系の数理」　朝倉書店　*2004*